AN850

Inductively Coupled Thermistor

Author: *Joseph Julicher*
Microchip Technology Inc.

INTRODUCTION

This application note describes a temperature sensor using a PICmicro® microcontroller. This temperature sensor is unique because it can work through non-ferromagnetic walls. Interfacing to a thermistor is normally a very straightforward task. Interfacing to a thermistor on the other side of a wall gets a little more complicated. Interfacing through walls can be a very valuable feature in certain applications, notably temperature sensing for refrigerators or sensing inside of a hazardous gas environment where an isolation barrier is a safety issue.

Implementation

Thermistors come in all types and values. For temperature sensing, most applications call for a NTC (negative temperature coefficient) type of thermistor where the resistance goes down with increasing temperature. Naturally, the resistance does not go down linearly, so some processing is required to translate the resistance to a temperature. This is of course a perfect job for a PICmicro microcontroller. The normal method of interfacing to a thermistor is shown below.

FIGURE 1: BASIC THERMISTOR CIRCUIT



This method is inappropriate for a through wall temperature system because it requires a wire to connect the thermistor to the PICmicro MCU. DC currents cannot be transmitted through a wall, so this method cannot be directly converted to a through wall system.

To sense through the wall we must get current flowing through the wall. This is easily done with a pair of coils of sufficient diameter to couple through the wall. About 100 winds of wire around a 12cm disk (size of CD) will provide sufficient inductance and size to couple through ½ inch. The PICmicro MCU can source up to 25 mA so that is how the coil will be energized. See the schematic below for the new circuit.

FIGURE 2: BLOCK DIAGRAM OF INDUCTIVELY COUPLED SENSOR



With each side of the transformer on different sides of the wall, we have created an isolation transformer. A thermistor is loading the secondary while the PICmicro MCU is driving the primary. Now the PICmicro MCU just needs a way to detect the voltage changes of the primary due to the loading on the secondary. One method is to feed a long pulse train to the coil and look at the amplitude modulation caused by the thermistor. This is effective, but it has two problems. Problem number one is, the continuous current through the thermistor will heat it and cause errors in the measurement. The second problem is, the circuitry to get a clean reading is complex and therefore not cheap.

A preferred method is to send a single pulse into the inductor and look at the transient response of the coil. We can then simplify the circuitry and eliminate the self-heating. By placing a capacitor and resistor across the primary, a RLC type circuit is formed (see Appendix B). This circuit will have a gentle peak when we energize it. See Figure 3.

FIGURE 3: RLC PULSE RESPONSE



The idea is to have the initial capacitor current dominate the first microsecond so the circuit does not look like the infinite resistance to the coil. As the capacitor charges, the coil current starts to become the dominant factor and the load becomes visible as a variation in coil peak voltage. The RC values must be adjusted to maintain a ratio of approximately 10000:1 with the inductor. With this configuration, a load on the secondary coil causes a change in the peak voltage on the primary. See Figure 4, set for the same scale as Figure 3.

FIGURE 4: INDUCTIVELY LOADED RLC PULSE RESPONSE



The peak voltage changes with the load voltage. This is a very useful feature. To make a good measurement without an amazingly fast analog-to-digital converter we can add a sample and hold circuit. Because the voltage variation is very small (about 500 mV in this example) we need a gain of about 10 to get a 5V range. The 2.75V offset needs to be removed so a difference amplifier is used to subtract the offset and then multiply the gain. Appendix B shows the test circuit schematic.

The calibration for this circuit is simple. Adjust the pot with no load on the secondary inductor. Adjust the pot until the analog-to-digital converter (ADC) is no longer reading a full-scale voltage. Any load on the secondary will cause the voltage to drop. The gain should be set to get a good reading over the desired range.

Theory of Operation

The PICmicro MCU sends a pulse to the inductor, which induces a voltage in the secondary coil. The secondary voltage across the thermistor causes a current, which is seen as a voltage drop, on the primary. The larger the secondary current, the larger the voltage drop at the primary. The first Op Amp, U1A, implements a high speed peak hold circuit by only passing current that charges the capacitor, but not allowing the capacitor to discharge. The second Op Amp, U1B, buffers the capacitor to the difference amplifier. This prevents the capacitor voltage from dropping too fast. The third Op Amp, U1C, subtracts the offset voltage and multiplies the difference by a gain of 10. The offset voltage is provided by the fourth Op Amp, U1D. The result is read by the ADC. The capacitor (C2) is drained between reads by an output from the PICmicro MCU.

The diode prevents the PICmicro MCU from charging the capacitor. By a small change in the software, this diode could be eliminated if the PICmicro MCU pin were left as an input pin at high impedance until the capacitor needed discharging. Alternatively, the pin would not be required at all if a suitable load resistor were provided for the capacitor. This resistor would have to be large enough that the capacitor did not drain too much before the ADC sample period passed and small enough to drain the capacitor between measurements. Here is the code for a PIC12C67X that takes a measurement.

```
measure          ; do the measurement
     bcf    INTCON,GIE   ; disable irq's
     bsf    GPIO,holdcap ; arm the cap
     bsf    GPIO,coil    ; charge the coil
     nop                 ; wait a bit
     bcf    GPIO,coil    ; Turn off coil
     bsf    ADCON0,GO    ; start ADC
     btfsc  ADCON0,GO    ; wait for ADC
     goto   $-1          ;
     bcf    GPIO,holdcap ; dump the cap
     bsf    INTCON,GIE   ; enable irq's
     movf   ADRES,W      ; result to W
     return              ; all done
```

The slowest part of the measurement is waiting for the ADC to finish. In the test system, GPIO4 was used to drive an LED with a PWM signal. This PWM was generated with a Timer0 interrupt. To prevent the Timer0 interrupt from affecting the pulse timing, all interrupts are disabled during the critical section of the measurement code. The PICmicro MCU is operating from its internal RC oscillator. This leaves a few pins to accomplish other tasks.

Figure 5 illustrates the complete circuit performance using a 10k pot in place of a thermistor. The offset was adjusted until the input to the ADC was 5V without the secondary coil in place and without clipping. With the coil in place, the resistor was swept over its entire range and produces values inside the gray area. With a suitable scaling table, this output could easily be converted to a resistance or a temperature.

Conclusion

Using inductive coupling is common with keyless entry, low frequency RF and power supplies. This application note shows that inductive pulse coupling can also be effectively used to transfer information, like temperature sensing, through a non-ferromagnetic barrier.

FIGURE 5: RANGE OF RESPONSE AT THE OUTPUT



APPENDIX A: CODE LISTING

```
;**********************************************************************
;                                                                     *
;     Filename:        fridge.asm                                     *
;     Date:            10/13/2000                                     *
;     File Version:    1.0                                            *
;                                                                     *
;     Author:          Joseph Julicher                                *
;     Company:         Microchip Technology                           *
;                                                                     *
;**********************************************************************

    list      p=12ce674             ; list directive to define processor
    #include <p12ce674.inc>         ; processor specific variable definitions
    errorlevel  -302                ; suppress message 302 from list file

    __CONFIG   _CP_OFF & _WDT_OFF & _MCLRE_OFF & _PWRTE_ON & _INTRC_OSC_NOCLKOUT


;***** VARIABLE DEFINITIONS
w_temp      EQU     0x70            ; variable used for context saving
status_temp EQU     0x71            ; variable used for context saving

tsr         EQU     0x72            ; transmit shift register
bitcount    EQU     0x73            ; transmit bit counter
led         EQU     0x74            ; LED brightness
counter     EQU     0x75            ; LED PWM counter
temp        EQU     0x76            ; holding for PWM status

;***** CONSTANTS DEFINITIONS
speed       EQU     0xDf            ; PWM period constant

;***** PIN DEFINITIONS
holdcap     EQU     0x02            ; GPIO pin for the hold cap
coil        EQU     0x01            ; GPIO pin for the coil
pwm         EQU     0x04            ; GPIO pin for the pwm (LED brightness)

;**********************************************************************
            ORG     0x000           ; processor reset vector
            goto    main            ; go to beginning of program


            ORG     0x004           ; interrupt vector location
            movwf   w_temp          ; save off current W register contents
            movf    STATUS,w        ; move status register into W register
            movwf   status_temp     ; save off contents of STATUS register

            incf    counter,w       ; PWM routine
            addwf   led,w
            btfss   STATUS,C
            bcf     GPIO,pwm
            btfsc   STATUS,C
```

```
            bsf     GPIO,pwm
            movwf   counter

            bcf     INTCON,T0IF     ; clear the TMR0 flag
            bsf     INTCON,T0IE     ; reenable TMR0 interrupt

            movlw   speed
            movwf   TMR0

            movf    status_temp,w   ; retrieve copy of STATUS register
            movwf   STATUS          ; restore pre-isr STATUS register contents
            movf    w_temp,w        ; restore W register
            retfie                  ; return from interrupt

main
         call    0x7FF              ; retrieve factory calibration value
         bsf     STATUS,RP0         ; set file register bank to 1
         movwf   OSCCAL             ; update register with factory cal value
         bcf     STATUS,RP0         ; set file register bank to 0

         clrf    TMR0               ; clear the timer
         clrf    counter
         clrf    led

; setup GPIO
         clrf    GPIO               ; set all I/O's to 0
         clrf    INTCON             ; clear all flags and enables
         bsf     INTCON,T0IE        ; enable TMR0 interrupt
         bsf     INTCON,GIE         ; enable all interrupts

         bsf     STATUS, RP0        ; Select Page 1
         clrf    OPTION_REG         ; clear all options
         bsf     OPTION_REG,NOT_GPPU; Turn off weak pullup

         movlw   B'00001001         ; GPIO 0 is Input
                                    ; GPIO 1 is Output
                                    ; GPIO 2 is Output
                                    ; GPIO 3 is Input
                                    ; GPIO 4 is Output
                                    ; GPIO 5 is Output
         movwf   TRISIO

         movlw   B'00000110         ; GP0 is analog, VREF is Vdd
         movwf   ADCON1             ; Configure A/D Inputs

         bcf     PIE1,ADIE          ; disable A/D Interrupts

         bcf     STATUS, RP0        ; Select Page 0
         movlw   B'01000001         ; 8 Tosc clock, A/D is on, Channel 0 is selected
         movwf   ADCON0             ;
         bcf     PIR1, ADIF         ; Clear A/D interrupt flag bit

repeat   call    measure            ; make a measurement
         movwf   led                ; set the LED brightness
         movlw   D'56               '; wait 200 loops or 1ms
delay    nop                        ;
         addlw   D'1                ;
         btfss   STATUS,Z           ;
         goto    delay              ;
         goto    repeat             ;

measure  ; do the measurement
         bcf     INTCON,GIE         ; disable all interrupts
         bsf     GPIO,holdcap       ; arm the cap
         bsf     GPIO,coil          ; charge the coil
         nop
         bcf     GPIO,coil          ; Turn off coil
```

```
                                ; wait for the inductor collapse to finish
        bsf     ADCON0,GO       ; start ADC
        btfsc   ADCON0,GO       ; wait for ADC to finish
        goto    $-1             ; go back if not finished yet
        bcf     GPIO,holdcap    ; dump the cap
        bsf     INTCON,GIE      ; enable all interrupts
        movf    ADRES,W         ; move the result to W
        return                  ; all done

        END
```

APPENDIX B: TEST SCHEMATIC

TEST SCHEMATIC



NOTES:

Note the following details of the code protection feature on PICmicro® MCUs.

- The PICmicro family meets the specifications contained in the Microchip Data Sheet.
- Microchip believes that its family of PICmicro microcontrollers is one of the most secure products of its kind on the market today, when used in the intended manner and under normal conditions.
- There are dishonest and possibly illegal methods used to breach the code protection feature. All of these methods, to our knowledge, require using the PICmicro microcontroller in a manner outside the operating specifications contained in the data sheet. The person doing so may be engaged in theft of intellectual property.
- Microchip is willing to work with the customer who is concerned about the integrity of their code.
- Neither Microchip nor any other semiconductor manufacturer can guarantee the security of their code. Code protection does not mean that we are guaranteeing the product as “unbreakable”.
- Code protection is constantly evolving. We at Microchip are committed to continuously improving the code protection features of our product.

If you have any further questions about this matter, please contact the local sales office nearest to you.

Printed on recycled paper.

DNV Certification, Inc. USA
DNV MSC The Netherlands Accredited by the RvA
ISO 9001 / QS-9000 REGISTERED FIRM

Microchip received QS-9000 quality system certification for its worldwide headquarters, design and wafer fabrication facilities in Chandler and Tempe, Arizona in July 1999 and Mountain View, California in March 2002. The Company’s quality system processes and procedures are QS-9000 compliant for its PICmicro® 8-bit MCUs, KEELOQ® code hopping devices, Serial EEPROMs, microperipherals, non-volatile memory and analog products. In addition, Microchip’s quality system for the design and manufacture of development systems is ISO 9001 certified.

WORLDWIDE SALES AND SERVICE

AMERICAS

Corporate Office
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7200 Fax: 480-792-7277
Technical Support: 480-792-7627
Web Address: http://www.microchip.com

Rocky Mountain
2355 West Chandler Blvd.
Chandler, AZ 85224-6199
Tel: 480-792-7966 Fax: 480-792-4338

Atlanta
500 Sugar Mill Road, Suite 200B
Atlanta, GA 30350
Tel: 770-640-0034 Fax: 770-640-0307

Boston
2 Lan Drive, Suite 120
Westford, MA 01886
Tel: 978-692-3848 Fax: 978-692-3821

Chicago
333 Pierce Road, Suite 180
Itasca, IL 60143
Tel: 630-285-0071 Fax: 630-285-0075

Dallas
4570 Westgrove Drive, Suite 160
Addison, TX 75001
Tel: 972-818-7423 Fax: 972-818-2924

Detroit
Tri-Atria Office Building
32255 Northwestern Highway, Suite 190
Farmington Hills, MI 48334
Tel: 248-538-2250 Fax: 248-538-2260

Kokomo
2767 S. Albright Road
Kokomo, Indiana 46902
Tel: 765-864-8360 Fax: 765-864-8387

Los Angeles
18201 Von Karman, Suite 1090
Irvine, CA 92612
Tel: 949-263-1888 Fax: 949-263-1338

New York
150 Motor Parkway, Suite 202
Hauppauge, NY 11788
Tel: 631-273-5305 Fax: 631-273-5335

San Jose
Microchip Technology Inc.
2107 North First Street, Suite 590
San Jose, CA 95131
Tel: 408-436-7950 Fax: 408-436-7955

Toronto
6285 Northam Drive, Suite 108
Mississauga, Ontario L4V 1X5, Canada
Tel: 905-673-0699 Fax: 905-673-6509

ASIA/PACIFIC

Australia
Microchip Technology Australia Pty Ltd
Suite 22, 41 Rawson Street
Epping 2121, NSW
Australia
Tel: 61-2-9868-6733 Fax: 61-2-9868-6755

China - Beijing
Microchip Technology Consulting (Shanghai) Co., Ltd., Beijing Liaison Office
Unit 915
Bei Hai Wan Tai Bldg.
No. 6 Chaoyangmen Beidajie
Beijing, 100027, No. China
Tel: 86-10-85282100 Fax: 86-10-85282104

China - Chengdu
Microchip Technology Consulting (Shanghai) Co., Ltd., Chengdu Liaison Office
Rm. 2401, 24th Floor,
Ming Xing Financial Tower
No. 88 TIDU Street
Chengdu 610016, China
Tel: 86-28-86766200 Fax: 86-28-86766599

China - Fuzhou
Microchip Technology Consulting (Shanghai) Co., Ltd., Fuzhou Liaison Office
Unit 28F, World Trade Plaza
No. 71 Wusi Road
Fuzhou 350001, China
Tel: 86-591-7503506 Fax: 86-591-7503521

China - Shanghai
Microchip Technology Consulting (Shanghai) Co., Ltd.
Room 701, Bldg. B
Far East International Plaza
No. 317 Xian Xia Road
Shanghai, 200051
Tel: 86-21-6275-5700 Fax: 86-21-6275-5060

China - Shenzhen
Microchip Technology Consulting (Shanghai) Co., Ltd., Shenzhen Liaison Office
Rm. 1315, 13/F, Shenzhen Kerry Centre,
Renminnan Lu
Shenzhen 518001, China
Tel: 86-755-2350361 Fax: 86-755-2366086

China - Hong Kong SAR
Microchip Technology Hongkong Ltd.
Unit 901-6, Tower 2, Metroplaza
223 Hing Fong Road
Kwai Fong, N.T., Hong Kong
Tel: 852-2401-1200 Fax: 852-2401-3431

India
Microchip Technology Inc.
India Liaison Office
Divyasree Chambers
1 Floor, Wing A (A3/A4)
No. 11, O'Shaugnessey Road
Bangalore, 560 025, India
Tel: 91-80-2290061 Fax: 91-80-2290062

Japan
Microchip Technology Japan K.K.
Benex S-1 6F
3-18-20, Shinyokohama
Kohoku-Ku, Yokohama-shi
Kanagawa, 222-0033, Japan
Tel: 81-45-471- 6166 Fax: 81-45-471-6122

Korea
Microchip Technology Korea
168-1, Youngbo Bldg. 3 Floor
Samsung-Dong, Kangnam-Ku
Seoul, Korea 135-882
Tel: 82-2-554-7200 Fax: 82-2-558-5934

Singapore
Microchip Technology Singapore Pte Ltd.
200 Middle Road
#07-02 Prime Centre
Singapore, 188980
Tel: 65-6334-8870 Fax: 65-6334-8850

Taiwan
Microchip Technology (Barbados) Inc.,
Taiwan Branch
11F-3, No. 207
Tung Hua North Road
Taipei, 105, Taiwan
Tel: 886-2-2717-7175 Fax: 886-2-2545-0139

EUROPE

Austria
Microchip Technology Austria GmbH
Durisolstrasse 2
A-4600 Wels
Austria
Tel: 43-7242-2244-399
Fax: 43-7242-2244-393

Denmark
Microchip Technology Nordic ApS
Regus Business Centre
Lautrup hoj 1-3
Ballerup DK-2750 Denmark
Tel: 45 4420 9895 Fax: 45 4420 9910

France
Microchip Technology SARL
Parc d'Activite du Moulin de Massy
43 Rue du Saule Trapu
Batiment A - ler Etage
91300 Massy, France
Tel: 33-1-69-53-63-20 Fax: 33-1-69-30-90-79

Germany
Microchip Technology GmbH
Gustav-Heinemann Ring 125
D-81739 Munich, Germany
Tel: 49-89-627-144 0 Fax: 49-89-627-144-44

Italy
Microchip Technology SRL
Centro Direzionale Colleoni
Palazzo Taurus 1 V. Le Colleoni 1
20041 Agrate Brianza
Milan, Italy
Tel: 39-039-65791-1 Fax: 39-039-6899883

United Kingdom
Microchip Ltd.
505 Eskdale Road
Winnersh Triangle
Wokingham
Berkshire, England RG41 5TU
Tel: 44 118 921 5869 Fax: 44-118 921-5820

05/16/02